A simple hydrologic mass balance model

C. C. Yen

Williamson and Schmid, 17782 Sky Park Boulevard, Irvine, CA 92714, USA

T. V. Hromadka II

Williamson and Schmid, 17782 Sky Park Boulevard, Irvine, CA 92714; Applied Mathematics Department, California State University, Fullerton, CA 92634, USA

A simple hydrologic mass balance model has been developed to study the time evolution of soil-water storage in an urban watershed. Advantages of using the mass balance model is that it only requires climatological data, it is easy to implement, and the model provides a synthetic record of long-term watershed behaviour. Water conservation can be analysed by varying the model parameters, such as infiltration rate, irrigation water, among other factors.

INTRODUCTION

Numerous mathematical models have been developed to simulate various hydrologic systems. An extensive literature review by Hromadka (1986) indicates that the simpler hydrologic models perform as good as or better than the more complex hydrologic models. In this paper, the focus is not upon runoff rates developed from storm events, but rather upon the time evolution of soil-water storage to the effects of rainfall, irrigation, evapotranspiration, and other factors. A simple hydrologic mass balance model is developed which accommodates several important mass-balance elements, using data which is readily obtained by field inspection or from local agency resources.

The hydrologic soil-water mass (water) balance model can be represented by the following mass balance equation:

$$I - 0 = \frac{\Delta s}{\Delta t} \tag{1}$$

where

I = inflow per unit time
0 = outflow per unit time
$\Delta s/\Delta t$ = the change in storage within the system per unit time

The uncertainties of the mass balance equation (equation (1)) lie mainly in the inability to properly measure or estimate the various inflow (I) and outflow (0) terms. Nevertheless, the mass balance model is a useful tool to simulate a long-term hydrologic system, such as the soil-water storage in a typical residential subdivision lot.

MODEL DESCRIPTIONS

A more generalized version of the mass balance equation will explain the various components of the hydrologic cycle and provide an insight to the problem-solving techniques. The hydrologic budget can be represented as

$$P + I_r - ET - I_t - I_f - S_d - R = \Delta s \tag{2}$$

where

P = precipitation per unit time
I_r = irrigation per unit time
ET = evapotranspiration per unit time
I_t = vegetation interception per unit time
I_f = infiltration per unit time
S_d = surface detention per unit time
R = surface runoff per unit time
Δs = change in storage per unit time

Precipitation submodel

In this submodel, the continuous rainfall record is used. If raingage data are unavailable within the study area, then nearby rainfall station records should be used with geographical adjustment.

Irrigation submodel

In the urban watershed, the irrigation water is limited to lawn and garden usages. By subtracting the averaged water usage per capita, per household, from the water meter record, the averaged irrigation water can be estimated for each household.

Evapotranspiration submodel

An equation for estimating evapotranspiration developed by the Agricultural Research Service (ARS) is used in this submodel. The evapotranspiration for any given day is determined as follows (Holtan *et al.*, 1975):

$$ET = GI \cdot k \cdot E_p \cdot \left(\frac{S - SA}{AWC}\right)^x \tag{3}$$

where

ET = evapotranspiration (in/day)
GI = growth index of crop in % of maturity

Received March 1988. Discussion closes June 1988.

Table 1. Hydrologic capacities of soil texture classes

Texture class	S (%)	G (%)	AWC (%)	x AWC/G
Coarse sand	24.4	17.7	6.7	0.38
Coarse sandy loam	24.5	15.8	8.7	0.55
Sand	32.3	19.0	13.3	0.70
Loamy sand	37.0	26.9	10.1	0.38
Loamy fine sand	32.6	27.2	5.4	0.20
Sandy loam	30.9	18.6	12.3	0.66
Fine sandy loam	36.6	23.5	13.1	0.56
Very fine sandy loam	32.7	21.0	11.7	0.56
Loam	30.0	14.4	15.6	1.08
Silt loam	31.3	11.4	19.9	1.74
Sandy clay loam	25.3	13.4	11.9	0.89
Clay loam	25.7	13.0	12.7	0.98
Silty clay loam	25.3	8.4	14.9	1.77
Sandy clay	19.4	11.6	7.8	0.67
Silty clay	21.4	9.1	12.3	1.34
Clay	18.8	7.3	11.5	1.58

S = total porosity − 15 bar moisture %, G = total porosity − 0.3 bar moisture %, and $AWC = S$ minus G

k = ratio of GI to pan evaporation, usually 1.0–1.2 for short grasses, 1.2–1.6 for crops up to shoulder height, and 1.6–2.0 for forest
E_p = pan evaporation (in./day)
S = total porosity of soil
SA = available porosity (unfilled by water)
X = AWC/G (G = moisture freely drained by gravity)

Equation (3) is used by the ARS in its USDAHL-74 model of watershed hydrology in combination with GI curves to calculate daily evapotranspiration. Representative values for S, G, and AWG are given in Table 1.

Vegetation interception submodel

The amount of water intercepted is a function of (1) the storm character, (2) amount of plant foliage and its character and orientation, and (3) the season of the year. For long-term hydrologic simulation, a constant interception storage that will be retained on the foliage against the forces of wind and gravity (usually varies between 0.01 and 0.05, Viessman *et al.*, 1977) is used in this submodel.

Infiltration submodel

The rate of infiltration is influenced by the type and extent of vegetal cover, the condition of the surface crust, temperature, rainfall intensity, physical properties of the soil, and water quality. The volume of storage available below ground is also a factor affecting infiltration rates. The ϕ-index model (Viessman *et al.*, 1977) is used in this submodel.

Surface detention submodel

Precipitation that reaches the ground may infiltrate, flow over the surface, or become trapped in numerous small depressions from which the only escape is evaporation or infiltration. In this model, the surface detention storage can be either increased to its maximum capacity by rainfall or decreased to zero by evaporation and/or infiltration.

Surface runoff submodel

In this model, the surface runoff is defined as the sum of excess precipitation and irrigation water. No flow routing is used in this submodel.

Storage submodel

Subsurface storage and groundwater storage are included in this submodel. The rules for interaction between subsurface storage and groundwater storage are as follows:

1. the subsurface storage has to be filled first,
2. after subsurface storage reaches a prescribed level, the water deep-percolates into the groundwater storage,
3. a portion of the groundwater storage is available for consumptive use through capillary soil-water flow when the subsurface storage is totally depleted, and
4. no subsurface water movement is included in this model.

MODEL USAGES

In general, the hydrologic mass balance (budget) model is a useful tool for simulating the long term soil-water storage evolution. A primary usage of this model is for analysis of water conservation. By investigating each submodel, different strategies can be studied to achieve a water conservation goal. Therefore, a thorough study of the entire hydrologic system should be conducted before the final planning is made. Finally, the economic, political, and aesthetic aspects should be incorporated in making the final planning.

DATA REQUIREMENTS

Hydrologic data are needed to describe precipitation, evaporation, transpiration, infiltration, and other components of the hydrologic system. Sources of data are numerous, with the US Geological Survey being a major source for streamflow and groundwater data. The National Weather Service (NOAA or National Oceanic and Atmospheric Administration) is the major collector of meteorologic data. Many other federal, state, and local agencies and other organizations also compile hydrologic data.

Precipitation data

Daily rainfall data and continuous rainfall data are best suited for the long term and short term hydrologic system simulation, respectively.

Irrigation data

In agricultural areas, water meter records can be used to determine the irrigation water consumption. In urban watersheds, the irrigation water consumption has to be interpolated for each household from its water meter record.

Pan evaporation and temperature data

Pan evaporation data is often scarce to obtain. Therefore, the temperature data can be used to correlate the available pan evaporation data to the study site.

Vegetation interception data

An area-averaged interception storage should be estimated for the study area by calculating the weighted average of different plant types and their foliage coverages.



Fig. 1. Location map for study area

Infiltration data

For long-term hydrologic system simulation, an area-averaged infiltration rate can be estimated from the S.C.S. hydrologic soil groups (US Bureau of Reclamation, 1973).

Surface detention data

Surface data can be obtained from field investigation.

Surface runoff and groundwater storage data

Either surface runoff data or groundwater data is need to calibrate the hydrologic mass balance model.

APPLICATION

A simple hydrologic mass balance model is developed for a small residential area in an urban watershed at the City of San Clemente, California (Fig. 1). All the surface waters are drained into the streets, and then carried into a storm drain system. The main concern is the slope stability of this hillside residential community. The instability of the slope is suspected to be due to the changing of the subsurface and groundwater storages after watershed urbanization.

Daily precipitation data are obtained from the OCEMA (Orange County Environmental Management Agency). Daily temperature data are obtained from the California Climatological Data. Monthly pan evaporation data are also obtained from OCEMA at the nearby station which is correlated to the daily temperature data to approximate the daily evapotranspiration value. No stream gage nor groundwater data are available. New groundwater observation wells have been installed in this area to record the change of subsurface and groundwater storages. When this information is available, the hydrologic mass balance model can be calibrated. An example model output is included hereafter to depict the model schematic and assumptions, and also illustrates the results of the computer code supplied in Appendix B.

REFERENCES

1 Holtan, H. N., Stilter, G. J., Henson, W. H. and Lopez, N. C. USDAHL-74 Revised Model of Watershed Hydrology, US Department of Agriculture, ARS Tech. Bulletin No. 1518, Washington DC, 1975
2 Hromadka II, T. V. The State-of-the-art in Hydrologic Models: A Review of Progress, Proceedings: ENVIROSOFT '86 Conference, Newport Beach, California, *Computational Mechanics Publications*, 1986
3 US Bureau of Reclamation, Design of Small Dams, P.544, US Government Printing Office, Washington DC, 1973
4 Viessman, W., Jr, Knapp, J. W., Lewis, G. L. and Harbaugh, T. E. *Introduction to Hydrology*, 2nd ed., IEP Publishers, New York, 1977

APPENDIX A

INPUT SEQUENCES AND OUTPUT RESULTS

```
SURFACE-SUBSURFACE WATER BUDGET MODEL
ENTER A [1] TO CREAT A NEW DATA FILE
   OR A [2] TO EXECUTE AN EXISTING FILE
   OR A [3] TO EXIST PROGRAM
=====> 1
ENTER THE FILE NAME THAT STORES THE INPUT DATA
=====> T1.DAT
ENTER TOTAL AREA (SQUARE FEET) = 5000
ENTER MAXIMUM SUBSURFACE STORAGE (CUBIC FEET) = 25000
ENTER INITIAL SUBSURFACE STORAGE (CUBIC FEET)
ALLOWABLE VALUE IS BETWEEN [0] AND [ 25000.00]
=====> 0
ENTER DRIVE WAY AND ROOF COVERAGE (%)
ALLOWABLE VALUE IS BETWEEN [0] AND [100]
=====> 45
ENTER TREE AND BUSH COVERAGE (%)
ALLOWABLE VALUE IS BETWEEN [0] AND [55.0]
```

```
=====> 5
ENTER DEPTH (INCHES) OF INTERCEPTION
=====> .1
ENTER CONCRETE WALK WAY AND PATIO COVERAGE (%)
ALLOWABLE VALUE IS BETWEEN [0] AND [50.0]
=====> 5
ENTER GRASS COVERAGE (%)
ALLOWABLE VALUE IS BETWEEN [0] AND [45.0]
=====> 25
IMPERVIOUS PONDING AREA
ENTER PONDING AREA(%) THAT DRAINS TO IMPERVIOUS AREA
ALLOWABLE VALUE IS BETWEEN [0] AND [20.0]
=====> 5
ENTER PONDING DEPTH (INCHES)
=====> 1
ENTER PONDING AREA(%) THAT DRAINS TO PERVIOUS AREA
ALLOWABLE VALUE IS BETWEEN [0] AND [15.0]
=====> 0
PERVIOUS PONDING AREA
ENTER PONDING AREA(%) THAT DRAINS TO IMPERVIOUS AREA
ALLOWABLE VALUE IS BETWEEN [0] AND [15.0]
=====> 5
ENTER PONDING DEPTH (INCHES)
=====> 1
ENTER PONDING AREA(%) THAT DRAINS TO PERVIOUS AREA
ALLOWABLE VALUE IS BETWEEN [0] AND [10.0]
=====> 0
ENTER AREA (%) SLOPE AWAY FROM HOUSE
ALLOWABLE VALUE IS BETWEEN [0] AND [10.0]
=====> 5
PERCENT OF BARE LAND AREA IS  5.0 %

T1.DAT          ...DATA FILE HAS BEEN CREATED
  SURFACE-SUBSURFACE WATER BUDGET MODEL
  ENTER A [1] TO CREAT A NEW DATA FILE
     OR A [2] TO EXECUTE AN EXISTING FILE
     OR A [3] TO EXIST PROGRAM
  =====> 2
  ENTER THE FILE NAME THAT STORES THE INPUT DATA
  =====> T1.DAT
  ENTER THE FILE NAME THAT STORES THE OUTPUT RESULTS
  =====> TEST.OUT
  INFILTRATION MODEL IS APPROXIMATED BY
        A CONSTANT SOIL LOSS RATE

  ENTER CONSTANT SOIL LOSS RATE (IN/HR) = .2
  ENTER PERCENT (%) OF GROUNDWATER STORAGE THAT
  CAN BE USED BY E.T. AFTER SUBSURAFCE STORAGE WAS
  DEPLETED BY E.T.
  =====> 30
  ENTER LONG TERM AVERAGE IRRIGATION WATER DEPHT (IN)
  =====> .5
  DO YOU WANT TO OUTPUT A DAILY WATER BUDGET FOR
  A USER SPECIFIED MONTH AND YEAR (Y/N) =====> N

 ==============================================================================
  ==> END OF WATER BUDGET SIMULATION <==

  NOTE: YOUR OUTPUT IS STORED IN [ TEST.OUT        ]
```




```
PERCENT OF IMPERVIOUS PONDING AREA THAT DRAINS TO IMPERVIOUS AREA =  5.0(%)
PERCENT OF IMPERVIOUS PONDING AREA THAT DRAINS TO PERVIOUS AREA =    .0(%)
PERCENT OF PERVIOUS PONDING AREA THAT DRAINS TO IMPERVIOUS AREA =  5.0(%)
PERCENT OF PERVIOUS PONDING AREA THAT DRAINS TO PERVIOUS AREA =    .0(%)
PERCENT OF AREA THAT SLOPES AWAY FROM THE HOUSE =  5.0(%)
==============================================================================
      NOTE : TOTAL AREA = 5000.00 SQUARE FEET
             MAXIMUM SUBSURFACE STORAGE =  25000.00 CUBIC FEET
             INITIAL SUBSURFACE STORAGE =       .00 CUBIC FEET
==============================================================================
```

MODEL ASSUMPTIONS

1. This model is based on the Water Budget Method, i.e.,

   RAINFALL - E.T. - INFILTRATION = CHANGE OF STORAGE

2. Rainfall is in inches/day.

3. E.T. is the sum of evaporation from pervious and impervious areas and transpiration from plants.

4. Infiltration includes infiltration into subsurface storage and deep percolation into groundwater storage.

5. Infiltration rate is approximated by the constant soil loss rate = 0.200 (in/hr).

6. The deep percolation occurs when the subsurface storage is greater than the maximum subsurface storage. The deep percolation volume is then set equal to volume which exceeds the maximum subsurface storage.

7. Storage includes subsurface and groundwater storages.

8. Subsurface storage is available for E.T. and part of the groundwater storage is availabe for E.T.

9. It is assumed that 30 percent of the groundwater storage is available for E.T. when subsurface storage is depleted.

10. It is assumed that the long term irrigation water which applies to the ponding and grass areas is equal to 0.5 inches/day.

11. It is assumed that the depth of interception for tree and bush is 0.1 inches.

12. Ponding depth of impervious area that drains to impervious area is 1.0 inches.

13. Ponding depth of impervious area that drains to pervious area is 0.0 inches.

14. Ponding depth of pervious area that drains to impervious area is 1.0 inches.

15. Ponding depth of pervious area that drains to pervious area is 0.0 inches.

1962 WATER BUDGET SUMMARY

	RAINFALL VOLUME (CU.FT.)	E.T. VOLUME (CU.FT.)	SURFACE/SUBSURFACE STORAGE (CU.FT.)	DEEP PERCOLATION (CU.FT.)	SURFACE RUNOFF (CU.FT.)	AVERAGED TEMPERATURE (°F)
JAN	NO RAINFALL DATA					
FEB	NO RAINFALL DATA					
MAR	NO RAINFALL DATA					
APR	NO RAINFALL DATA					
MAY	NO RAINFALL DATA					
JUN	NO RAINFALL DATA					
JUL	NO RAINFALL DATA					
AUG	NO RAINFALL DATA					
SEP	.00	464.86	20.83/ 1535.77	.00	166.04	73.0 °F
OCT	4.17	476.93	20.83/ 3093.00	.00	230.42	70.3 °F
NOV	29.17	488.71	20.83/ 4579.71	.00	241.25	67.0 °F
DEC	8.33	421.65	20.83/ 6156.18	.00	270.62	64.1 °F
SUM	41.67	1852.15	20.83/ 6156.18	.00	908.33	
AVG.						

****

1963 WATER BUDGET SUMMARY

	RAINFALL VOLUME (CU.FT.)	E.T. VOLUME (CU.FT.)	SURFACE/SUBSURFACE STORAGE (CU.FT.)	DEEP PERCOLATION (CU.FT.)	SURFACE RUNOFF (CU.FT.)	AVERAGED TEMPERATURE (°F)
JAN	83.33	550.92	22.92/ 7659.39	.00	287.55	62.6 °F
FEB	1195.83	1341.78	20.83/ 8863.60	.00	693.58	67.8 °F
MAR	879.17	2576.36	20.83/ 9050.79	.00	376.04	65.5 °F
APR	387.50	3048.83	1.01/ 8526.87	.00	69.92	65.0 °F
MAY	25.00	3279.03	.00/ 7534.27	.00	.00	68.0 °F
JUN	8.33	2633.69	2.13/ 7094.28	.00	.00	69.8 °F
JUL	.00	1679.49	20.83/ 7541.15	.00	115.35	72.9 °F
AUG	.00	1546.47	20.83/ 8107.20	.00	147.89	76.1 °F
SEP	1208.33	1603.54	20.83/ 9252.53	.00	646.96	81.8 °F
OCT	145.83	1226.39	20.83/10198.45	.00	233.92	74.7 °F
NOV	2291.67	1166.77	20.83/12212.60	.00	1298.25	69.0 °F
DEC	.00	878.54	20.83/13374.79	.00	219.69	68.2 °F
SUM	6225.00	21531.81	20.83/13374.79	.00	4089.15	
AVG.	518.75	1794.32		.00	340.76	70.1 °F

****

.
.
.

1984 WATER BUDGET SUMMARY

	RAINFALL VOLUME (CU.FT.)	E.T. VOLUME (CU.FT.)	SURFACE/SUBSURFACE STORAGE (CU.FT.)	DEEP PERCOLATION (CU.FT.)	SURFACE RUNOFF (CU.FT.)	AVERAGED TEMPERATURE (°F)
JAN	104.17	1160.81	20.83/16765.26	.00	225.14	70.3 °F
FEB	62.50	2929.75	.00/16033.42	.00	.00	69.8 °F
MAR	62.50	4132.15	.00/14223.85	.00	.34	72.1 °F
APR	220.83	3249.73	.00/13338.27	.00	44.18	70.8 °F
MAY	.00	5042.31	.00/10556.39	.00	.00	75.3 °F
JUN	12.50	3367.30	.00/ 9389.09	.00	.00	74.3 °F
JUL	.00	2106.67	20.83/ 9462.27	.00	59.74	81.8 °F
AUG	70.83	1881.97	20.83/ 9797.46	.00	114.08	81.0 °F
SEP	204.17	1798.39	20.83/10229.31	.00	161.43	84.5 °F
OCT	220.83	1390.48	20.83/11076.99	.00	243.10	75.0 °F
NOV	679.17	1350.04	20.83/12193.64	.00	399.97	69.0 °F
DEC	2162.50	1202.95	20.83/14278.11	.00	1135.50	63.9 °F
SUM	3800.00	29612.55	20.83/14278.11	.00	2383.48	
AVG.	316.67	2467.71		.00	198.62	74.0 °F

****

1985.WATER BUDGET SUMMARY

	RAINFALL VOLUME (CU.FT.)	E.T. VOLUME (CU.FT.)	SURFACE/SUBSURFACE STORAGE (CU.FT.)	DEEP PERCOLATION (CU.FT.)	SURFACE RUNOFF (CU.FT.)	AVERAGED TEMPERATURE (°F)
JAN	337.50	949.41	20.83/15580.57	.00	346.04	64.5 °F
FEB	683.33	2350.03	20.83/15667.49	.00	288.06	64.9 °F
MAR	245.83	3712.55	.00/14445.60	.00	36.43	63.2 °F
APR	50.00	4048.61	.00/12634.49	.00	.00	70.2 °F
MAY	8.33	4512.31	.00/10390.93	.00	.00	68.9 °F
JUN	.00	3719.90	.00/ 8858.53	.00	.00	73.4 °F

==> END OF WATER BUDGET SIMULATION <==

APPENDIX B

COMPUTER CODE

```
C  THIS IS A SURFACE-SUBSURFACE WATER BUDGET PROGRAM
C
      CHARACTER IMON(12)*3,KO*1,IFILE*15,OFILE*15
      DIMENSION IDAY(12),GI(12)
      DIMENSION TEMP(31),RAIN(31),EVP(12)
      DATA PR/0./,PT/0./,PIG/0./,PG/0./,PP/0./,PB/0./,
     C     PI/0./,PS/0./,PI1/0./,PP1/0./
      DATA IMON/'JAN','FEB','MAR','APR','MAY','JUN',
     C          'JUL','AUG','SEP','OCT','NOV','DEC'/
      DATA IDAY/31,28,31,30,31,30,31,31,30,31,30,31/
      DATA GI/.4,.8,1.,1.,1.,.8,.4,.4,.4,.4,.4,.4/
      DATA KO/'Y'/
      DATA IYEAR/1962/,IEND/1987/
      DATA XDI/0./,XDPI/0./,XDPI1/0./,XDPP/0./,XDPP1/0./,VOLSUR/0./,
     C     TTO/0./,TET/0./,TSUB/0./,TSUR/0./,TPER/0./,TRUN/0./
C
C  OPEN FILES
C
      OPEN(UNIT=1,FILE='TEMP.FIL',STATUS='OLD')
      OPEN(UNIT=2,FILE='RAIN.FIL',STATUS='OLD')
      OPEN(UNIT=3,FILE='EVAP.FIL',STATUS='OLD')
C
C  INPUT SUBAREA INFORMATION
C
 35   WRITE(*,*)'  SURFACE-SUBSURFACE WATER BUDGET MODEL'
      WRITE(*,*)'  ENTER A [1] TO CREAT A NEW DATA FILE'
      WRITE(*,*)'     OR A [2] TO EXECUTE AN EXISTING FILE'
      WRITE(*,*)'     OR A [3] TO EXIST PROGRAM'
      WRITE(*,4)
      READ(*,*,ERR=35)JKODE
      IF(JKODE.EQ.1)GO TO 37
      IF(JKODE.EQ.2)GO TO 90
      IF(JKODE.EQ.3)STOP '  '
      GO TO 35
 37   WRITE(*,*)'  ENTER THE FILE NAME THAT STORES THE INPUT DATA'
      WRITE(*,4)
      READ(*,34)IFILE
 34   FORMAT(A15)
      OPEN(UNIT=4,FILE=IFILE,STATUS='NEW')
 91   WRITE(*,1)
 1    FORMAT('   ENTER TOTAL AREA (SQUARE FEET) = ',\)
      READ(*,*,ERR=91)AREA
 92   WRITE(*,2)
 2    FORMAT('   ENTER MAXIMUM SUBSURFACE STORAGE (CUBIC FEET) = ',\)
      READ(*,*,ERR=92)SUBMAX
 25   WRITE(*,21)SUBMAX
 21   FORMAT('   ENTER INITIAL SUBSURFACE STORAGE (CUBIC FEET)',/,
     1 '   ALLOWABLE VALUE IS BETWEEN [0] AND [',F9.2,']')
      WRITE(*,4)
      READ(*,*,ERR=25)VOLSUB
      IF(VOLSUB.LT.0. .OR. VOLSUB.GT.SUBMAX)GO TO 25
 20   WRITE(*,3)
 3    FORMAT('   ENTER DRIVE WAY AND ROOF COVERAGE (%)',/,
     1 '   ALLOWABLE VALUE IS BETWEEN [0] AND [100]')
      WRITE(*,4)
```

```
4     FORMAT('   =====> ',\)
      READ(*,*,ERR=20)PR
      IF(PR.LT.0. .OR. PR.GT.100.)GO TO 20
      XP=100.-PR
      IF(XP.EQ.0.)GO TO 60
30    WRITE(*,5)XP
5     FORMAT('   ENTER TREE AND BUSH COVERAGE (%)',/,
     1 '   ALLOWABLE VALUE IS BETWEEN [0] AND [',F4.1,']')
      WRITE(*,4)
      READ(*,*,ERR=30)PT
      IF(PT.LT.0. .OR. PT.GT.XP)GO TO 30
93    WRITE(*,*)'  ENTER DEPTH (INCHES) OF INTERCEPTION'
      WRITE(*,4)
      READ(*,*,ERR=93)DINT
      XP=XP-PT
      IF(XP.EQ.0.)GO TO 60
40    WRITE(*,6)XP
6     FORMAT('   ENTER CONCRETE WALK WAY AND PATIO COVERAGE (%)',/,
     1 '   ALLOWABLE VALUE IS BETWEEN [0] AND [',F4.1,']')
      WRITE(*,4)
      READ(*,*,ERR=40)PIG
      IF(PIG.LT.0. .OR. PIG.GT.XP)GO TO 40
      XP=XP-PIG
      IF(XP.EQ.0.)GO TO 60
50    WRITE(*,7)XP
7     FORMAT('   ENTER GRASS COVERAGE (%)',/,
     1 '   ALLOWABLE VALUE IS BETWEEN [0] AND [',F4.1,']')
      WRITE(*,4)
      READ(*,*,ERR=50)PG
      IF(PG.LT.0. .OR. PG.GT.XP)GO TO 50
      XP=XP-PG
      IF(XP.EQ.0.)GO TO 60
55    WRITE(*,*)'  IMPERVIOUS PONDING AREA'
      WRITE(*,31)XP
31    FORMAT('   ENTER PONDING AREA(%) THAT DRAINS TO IMPERVIOUS ',
     1 'AREA',/,'   ALLOWABLE VALUE IS BETWEEN [0] AND [',F4.1,']')
      WRITE(*,4)
      READ(*,*,ERR=55)PI1
      IF(PI1.LT.0. .OR. PI1.GT.XP)GO TO 55
      DPI1=0.
      IF(PI1.EQ.0.)GO TO 65
94    WRITE(*,*)'  ENTER PONDING DEPTH (INCHES)'
      WRITE(*,4)
      READ(*,*,ERR=94)DPI1
      XP=XP-PI1
      IF(XP.EQ.0.)GO TO 60
65    WRITE(*,32)XP
32    FORMAT('   ENTER PONDING AREA(%) THAT DRAINS TO PERVIOUS AREA',
     1 /,'   ALLOWABLE VALUE IS BETWEEN [0] AND [',F4.1,']')
      WRITE(*,4)
      READ(*,*,ERR=65)PI
      IF(PI.LT.0. .OR. PI.GT.XP)GO TO 65
      DPI=0.
      IF(PI.EQ.0.)GO TO 56
96    WRITE(*,*)'  ENTER PONDING DEPTH (INCHES)'
      WRITE(*,4)
      READ(*,*,ERR=96)DPI
      XP=XP-PI
      IF(XP.EQ.0.)GO TO 60
```

```
56    WRITE(*,*)'  PERVIOUS PONDING AREA'
      WRITE(*,31)XP
      WRITE(*,4)
      READ(*,*,ERR=56)PP1
      IF(PP1.LT.0. .OR. PP1.GT.XP)GO TO 56
      DPP1=0.
      IF(PP1.EQ.0.)GO TO 66
97    WRITE(*,*)'  ENTER PONDING DEPTH (INCHES)'
      WRITE(*,4)
      READ(*,*,ERR=97)DPP1
      XP=XP-PP1
      IF(XP.EQ.0.)GO TO 60
66    WRITE(*,32)XP
      WRITE(*,4)
      READ(*,*,ERR=66)PP
      IF(PP.LT.0. .OR. PP.GT.XP)GO TO 66
      DPP=0.
      IF(PP.EQ.0.)GO TO 67
98    WRITE(*,*)'  ENTER PONDING DEPTH (INCHES)'
      WRITE(*,4)
      READ(*,*,ERR=98)DPP
67    XP=XP-PP
      IF(XP.EQ.0.)GO TO 60
70    WRITE(*,33)XP
33    FORMAT('   ENTER AREA (%) SLOPE AWAY FROM HOUSE',/,
     1 '   ALLOWABLE VALUE IS BETWEEN [0] AND [',F4.1,']')
      WRITE(*,4)
      READ(*,*,ERR=70)PS
      IF(PS.LT.0. .OR. PS.GT.XP)GO TO 70
      PB=XP-PS
      IF(PB.NE.0.)WRITE(*,8)PB
8     FORMAT('   PERCENT OF BARE LAND AREA IS ',F4.1,' %',/)
C
C  SAVE THE INPUT FILE
C
60    IF(JKODE.NE.1)GO TO 61
      WRITE(4,*)AREA,SUBMAX,VOLSUB
      WRITE(4,*)PR,PT,DINT,PIG
      WRITE(4,*)PG,PI1,DPI1,PI,DPI,PP1,DPP1,PP,DPP,PS,PB
      WRITE(*,*)'  '
      WRITE(*,46)IFILE
46    FORMAT(/,2X,A15,'...DATA FILE HAS BEEN CREATED',/)
      CLOSE (UNIT=4)
      GO TO 35
C
C  READ DATA FORM INPUT FILE
C
90    WRITE(*,*)'  ENTER THE FILE NAME THAT STORES THE INPUT DATA'
      WRITE(*,4)
      READ(*,34)IFILE
      WRITE(*,*)'  ENTER THE FILE NAME THAT STORES THE OUTPUT RESULTS'
      WRITE(*,4)
      READ(*,34)OFILE
      OPEN(UNIT=4,FILE=IFILE,STATUS='OLD')
      READ(4,*)AREA,SUBMAX,VOLSUB
      READ(4,*)PR,PT,DINT,PIG
      READ(4,*)PG,PI1,DPI1,PI,DPI,PP1,DPP1,PP,DPP,PS,PB
      CLOSE (UNIT=4)
C
```

```
C  ENTER MODEL VARIABLES
C
 61   WRITE(*,*)'  INFILTRATION MODEL IS APPROXIMATED BY'
      WRITE(*,*)'          A CONSTANT SOIL LOSS RATE '
      WRITE(*,15)
 15   FORMAT(/,'   ENTER CONSTANT SOIL LOSS RATE (IN/HR) = ',\)
      READ(*,*,ERR=60)FO
      DINF=FO*24.
 62   WRITE(*,*)'  ENTER PERCENT (%) OF GROUNDWATER STORAGE THAT'
      WRITE(*,*)'  CAN BE USED BY E.T. AFTER SUBSURAFCE STORAGE WAS'
      WRITE(*,*)'  DEPLETED BY E.T.'
      WRITE(*,4)
      READ(*,*,ERR=62)PGR
      IF(PGR.LT.0. .OR. PGR.GT.100.)GO TO 62
 991  WRITE(*,*)'  ENTER LONG TERM AVERAGE IRRIGATION WATER DEPHT (IN)'
      WRITE(*,4)
      READ(*,*,ERR=991)DIRR
      WRITE(*,27)
 27   FORMAT('   DO YOU WANT TO OUTPUT A DAILY WATER BUDGET FOR',/,
     1 '   A USER SPECIFIED MONTH AND YEAR (Y/N) =====> ',\)
      READ(*,23)KY
 23   FORMAT(A1)
      IF(KY.NE.KO)GO TO 80
 75   WRITE(*,24)
 24   FORMAT('   ENTER SPECIFIED MONTH BETWEEN [1] AND [12]')
      WRITE(*,4)
      READ(*,*,ERR=75)ISM
      IF(ISM.LT.1 .OR. LSM.GT.12)GO TO 75
 85   WRITE(*,26)IYEAR,IEND
 26   FORMAT('   ENTER SPECIFIED YEAR BETWEEN [',I4,'] AND [',I4,']')
      WRITE(*,4)
      READ(*,*,ERR=85)ISY
      IF(ISY.LT.IYEAR .OR. ISY.GT.IEND)GO TO 85
C..COMPOSED RATIO OF GROWTH INDEX TO PAN EVAPORATION
 80   XK=0.
      DA=PT+PG+PS+PI+PP
      DB=PT*1.5+(PS+PI+PP)*1.2+PG
      IF(DA.NE.0.)XK=DB/DA
C
C  OUTPUT SUBAREA INFORMATION
C
      OPEN(UNIT=7,FILE=OFILE,STATUS='UNKNOWN')
      WRITE(7,9)PR,PT,PIG
 9    FORMAT(3(/),
     1 5X,'                               RAINFALL',//,
     2 2X,35(' |'),//,3X,34(' |'),//,
     3 13X,'^',/,13X,'|',/,13X,'|',/,6X,'  EVAPORATION',/,13X,'|',/,
     3 5X,' -----------------',/,
     3 5X,' | DRIVE-WAY AND |',/,
     4 5X,' |  ROOF COVER   |',/,
     5 5X,' |',3X,F4.1,' %       |       ---------------',14X,'^',/,
     6 6X,17('-'),'       | TREE COVER |',14X,'|',/,
     7 13X,'|                     |',3X,F4.1,' %   |',9X,'EVAPORATION',/,
     8 6X,'        |                  ---------------',14X,'|',/,
     9 6X,'        |                             |',11X,21('-'),/,
     1 6X,'        |',21X,'|            | CONCRETE WALK WAY |',/,
     2 6X,'        |',21X,'|            |  AND PATIO COVER  |',/,
     3 6X,'        |',9X,' ',11X,'|      ^     |',7X,F4.1,' %        |',/,
     4 6X,'        |',9X,'|',11X,'|      |     ---------------------',/,
```

```
     5 6X,'       |         |          |   EVAPO -           |',/,
     6 6X,'       |    EVAPORATION     | TRANSPIRATION       |',/,
     7 6X,'       |          |          V      |              V')
       WRITE(7,10)PG,PB
 10    FORMAT(
     8 13X,'|     -----------   ---------------------------------',/,
     9 13X,'|     |  BARE |    |         GRASS COVER           |',/,
     1 13X,'|     |  LAND |    |',12X,F4.1,' %          |',/,
     2 13X,'|     | ',F4.1,' % |    --------------------------------',/,
     3 6X,'       |     -----------          |                ^',/,
     4 6X,'       |       ^    |             |                |',/,
     6 6X,'       |       |    |             |                |',/,
     7 3X,10('-'),'V-------|----V',13('-'),'V',16('-'),'|',14('-'),/,
     8 6X,'               |                                   |',/,
     9 6X,'               |      SURFACE    RUNOFF           |',/,
     1 6X,'               |                                   |',/,
     2 3X,18('-'),'|-----------------------------------|',14('-'),/,
     3 6X,'               V                                   V',/,
     4 6X,'                      SUBSURFACE STORAGE',//,
     2 3X,18('-'),'|-----------------------------------|',14('-'),/,
     2 21X,'|',35X,'|',/,
     3 6X,'               V                                   V',/,
     4 6X,'                      DEEP   PERCOLATION',//,1X,78('.'))
      WRITE(7,51)PI1,PI,PP1,PP,PS,AREA,SUBMAX,VOLSUB
 51   FORMAT('  PERCENT OF',
     8 ' IMPERVIOUS PONDING AREA THAT DRAINS TO IMPERVIOUS AREA = ',
     8 F4.1,'(%)',/,'  PERCENT OF IMPERVIOUS PONDING AREA THAT',
     8 ' DRAINS TO PERVIOUS AREA = ',F4.1,'(%)',/,'  PERCENT OF',
     8 ' PERVIOUS PONDING AREA THAT DRAINS TO IMPERVIOUS AREA = ',
     8 F4.1,'(%)',/,'  PERCENT OF PERVIOUS PONDING AREA THAT',
     8 ' DRAINS TO PERVIOUS AREA = ',F4.1,'(%)',/,
     8 '  PERCENT OF AREA THAT SLOPES AWAY FROM THE HOUSE',
     8 ' = ',F4.1,'(%)',/,1X,78('='),/,8X,
     1 'NOTE : TOTAL AREA = ',F7.2,' SQUARE FEET',/,15X,'MAXIMUM ',
     7 'SUBSURFACE STORAGE = ',F9.2,' CUBIC FEET',/,15X,'INITIAL ',
     8 'SUBSURFACE STORAGE = ',F9.2,' CUBIC FEET',/,1X,78('='))
      WRITE(7,19)FO,PGR,DIRR
 19   FORMAT(6(/),2X,'MODEL ASSUMPTIONS',/,1X,'=================',//,
     1 4X,'1.  This model is based on the Water Budget Method, i.e.,',
     2 //,8X,'RAINFALL - E.T. - INFILTRATION = CHANGE OF',
     3 ' STORAGE',//,4X,'2.  Rainfall is in inches/day.',//,
     4 4X,'3.  E.T. is the sum of evaporation from pervious and'
     5.,/,8X,'impervious areas and transpiration from plants.',//,
     6 4X,'4.  Infiltration includes infiltration into subsurface ',
     7 'storage',/,8X,'and deep percolation into groundwater storage.',
     2 //,4X,'5.  Infiltration rate is approximated by',/,
     3 8X,'the constant soil loss rate = ',F5.3,' (in/hr).',//,
     9 4X,'6.  The deep percolation occurs when the subsurface storage'
     1 ,/,8X,'is greater than the maximum subsurface storage. The deep',
     1 /,8X,'percolation volume is then set equal to volume which',
     1 /,8X,'exceeds the maximum subsurface storage.'
     7 //,4X,'7.  Storage includes subsurface and groundwater storages.'
     8 ,//,4X,'8.  Subsurface storage is available for E.T. and',
     9 /,8X,'part of the groundwater storage is availabe for E.T.',//,4X,
     1 '9.  It is assumed that ',F3.0,' percent of the groundwater',
     1 ' storage',/,8X,'is available for E.T. when subsurface storage',
     1 ' is depleted.',//,3X,'10.  It is assumed that the long term',
     1 ' irrigation water which applies',/,8X,'to the ponding and',
     1 ' grass areas is equal to ',F3.1,' inches/day.',/)
      WRITE(7,47)DINT,DPI1,DPI,DPP1,DPP
```

```
 47   FORMAT(3X,'11.  It is assumed that the depth of interception for',
     1 ' tree and bush',/,8X,'is ',F3.1,' inches.',//,3X,'12.  Ponding',
     2 ' depth of impervious area that drains to impervious area',/,8X,
     3 'is ',F3.1,' inches.',//,3X,'13.  Ponding',
     2 ' depth of impervious area that drains to pervious area',/,8X,
     3 'is ',F3.1,' inches.',//,3X,'14.  Ponding',
     2 ' depth of pervious area that drains to impervious area',/,8X,
     3 'is ',F3.1,' inches.',//,3X,'15.  Ponding',
     2 ' depth of pervious area that drains to pervious area',/,8X,
     3 'is ',F3.1,' inches.',6(/))
C
C   YEARLY WATER BUDGET SIMULATION LOOP
C
      DO 1000 II=IYEAR,IEND
      WRITE(7,11)II
 11   FORMAT(///,5X,I4,' WATER BUDGET SUMMARY',//,8X,'RAINFALL',3X,
     1 'E.T.  SURFACE/SUBSURFACE',5X,'DEEP',5X,'SURFACE',3X,'AVERAGED',
     1 /,9X,'VOLUME',3X,'VOLUME',7X,'STORAGE',7X,'PERCOLATION',2X,
     2 'RUNOFF',2X,'TEMPERATURE',/,8X,'(CU.FT.) (CU.FT.)      (CU.FT.)',
     3 8X,'(CU.FT.)',2X,'(CU.FT.)',5X,'(°F)',/,7X,71('-')/)
      YPER=0.
      YTO=0.
      YET=0.
      YRUN=0.
      YTEMP=0.
      READ(3,*,END=9999)(EVP(I),I=1,12)
C
C   MONTHLY WATER BUDGET SIMULATION LOOP
C
      DO 1100 JJ=1,12
      IF(II.EQ.IYEAR .AND. JJ.LT.9)GO TO 1200
      VOLTO=0.
      VOLET=0.
      VOLRUN=0.
      VOLPER=0.
      TTEMP=0.
      IDAY(2)=28
      IF(JJ.EQ.2 .AND. MOD(II,4).EQ.0)IDAY(2)=29
C
C  INPUT MONTHLY TEMPERATURE AND RAINFALL DATA
C
      READ(1,*,END=9999)(TEMP(I),I=1,IDAY(JJ))
      READ(2,*,END=9999)(RAIN(I),I=1,IDAY(JJ))
C
C  DAILY WATER BUDGET SIMULATION LOOP
C
      DO 110 I=1,IDAY(JJ)
 110  TTEMP=TTEMP+TEMP(I)
      XXK=XK*GI(JJ)
      DO 100 I=1,IDAY(JJ)
      ET=EVP(JJ)
      DEPTH=RAIN(I)
      VOLTO=VOLTO+DEPTH*AREA/12.
C..ESTIMATION OF ET AND PET
      TFRA=TEMP(I)/TTEMP
      ET=ET*TFRA
      VOLABC=VOLSUB+(PP+PS+PP1)*DEPTH*AREA*.01/12.
      PET=XXK*ET*VOLABC/SUBMAX/.115
```

```
C..ROOF COVER
      IF(PR.EQ.0.)GO TO 210
      A=AREA*PR*.01/12.
      IF(DEPTH.LT.ET)GO TO 200
      VOLRUN=VOLRUN+(DEPTH-ET)*A
      VOLET=VOLET+ET*A
      GO TO 210
 200  VOLET=VOLET+DEPTH*A
 210  CONTINUE
C..CONCRETE WALK WAY AND PATIO COVER
      IF(PIG.EQ.0.)GO TO 250
      A=AREA*PIG*.01/12.
      DIG=0.
      IF(DEPTH.LE.ET)GO TO 220
      VOLET=VOLET+ET*A
      DIG=DEPTH-ET
      GO TO 230
 220  VOLET=VOLET+DEPTH*A
C..TREE AND BUSH COVERAGE
 230  IF(PT.EQ.0.)GO TO 240
      DIT=0.
      A=AREA*PT*.01/12.
      XDI=XDI+DEPTH
      IF(XDI.GT.ET)GO TO 235
      VOLET=VOLET+XDI*A
      XDI=0.
      GO TO 240
 235  VOLET=VOLET+ET*A
      XDI=XDI-ET
      IF(XDI.LT.DINT)GO TO 240
      DIT=XDI-DINT
      XDI=DINT
C..IMPERVIOUS PONDING AREA
C....OVERLAND FLOW TO IMPERVIOUS AREA
 240  IF(PI1.EQ.0.)GO TO 245
      A=AREA*PI1*.01/12.
      XDPI1=XDPI1+DEPTH+DIRR
      XET=AMAX1(ET,PET)
      IF(XDPI1.GT.XET)GO TO 243
      VOLET=VOLET+XDPI1*A
      XDPI1=0.
      GO TO 245
 243  VOLET=VOLET+XET*A
      XDPI1=XDPI1-XET
      IF(XDPI1.LT.DPI1)GO TO 245
      VOLRUN=VOLRUN+(XDPI1-DPI1)*A
      XDPI1=DPI1
C....OVERLAND FLOW TO PERVIOUS AREA
 245  IF(PI.EQ.0.)GO TO 255
      A=AREA*PI*.01/12.
      XDPI=XDPI+DEPTH+DIRR
      XET=AMAX1(ET,PET)
      DPIP=0.
      IF(XDPI.GT.XET)GO TO 253
      VOLET=VOLET+XDPI*A
      XDPI=0.
      GO TO 255
 253  VOLET=VOLET+XET*A
      XDPI=XDPI-XET
      IF(XDPI.LT.DPI)GO TO 255
```

```
      DPIP=XDPI-DPI
      XDPI=DPI
C..PERVIOUS PONDING AREA
C....OVERLAND FLOW TO IMPERVIOUS AREA
 255  IF(PP1.EQ.0.)GO TO 257
      A=AREA*PP1*.01/12.
      XDPP1=XDPP1+DEPTH+DIRR
      XET=AMAX1(ET,PET)
      CALL BUDGET(XDPP1,XET,A,SUBMAX,VOLSUB,VOLPER,DINF,DPP1,
     C           DG1,VOLET,VOLRUN,PGR)
      VOLRUN=VOLRUN+DG1*A
C....OVERLAND FLOW TO PERVIOUS AREA
 257  IF(PP.EQ.0.)GO TO 250
      A=AREA*PP*.01/12.
      XDPP=XDPP+DEPTH+DIRR
      XET=AMAX1(ET,PET)
      CALL BUDGET(XDPP,XET,A,SUBMAX,VOLSUB,VOLPER,DINF,DPP,
     C           DG,VOLET,VOLRUN,PGR)
C..GRASS COVER
 250  IF(PG.EQ.0.)GO TO 260
      DAVE=((DEPTH+DIRR)*PG+DIT*PT+DG*PP+DIG*PIG+DPIP*PI)/PG
      A=AREA*PG*.01/12.
      CALL BUDGET(DAVE,PET,A,SUBMAX,VOLSUB,VOLPER,DINF,0.,
     C           DUM,VOLET,VOLRUN,PGR)
C..BARE LAND
 260  IF(PB.EQ.0.)GO TO 270
      A=AREA*PB*.01/12.
      CALL BUDGET(DEPTH,ET,A,SUBMAX,VOLSUB,VOLPER,DINF,0.,
     C           DUM,VOLET,VOLRUN,PGR)
C..AREA SLOPE AWAY FROM HOUSE
 270  A=AREA*PS*.01/12.
      CALL BUDGET(DEPTH,PET,A,SUBMAX,VOLSUB,VOLPER,DINF,0.,
     C           DUM,VOLET,VOLRUN,PGR)
C..CALCULATE SURFACE PONDING VOLUME
      VOLSUR=(XDI*PT+XDPI*PI+XDPI1*PI1+XDPP*PP+XDPP1*PP1)*AREA*.01/12.
C..END OF DAILY WATER BUDGET
C..SUMMARY OF DAILY WATER BUDGET
      IF(JJ.NE.ISM .OR. II.NE.ISY)GO TO 100
      IF(I.EQ.1)GO TO 400
      DTO=VOLTO-TTO
      DET=VOLET-TET
      DSUB=VOLSUB-TSUB
      DRUN=VOLRUN-TRUN
      DPER=VOLPER-TPER
      DSUB=VOLSUB-TSUB+DPER
      DSUR=VOLSUR-TSUR
C..OUTPUT WATER BUDGET
 400  WRITE(7,211)I,VOLTO,VOLET,VOLSUB,VOLPER,VOLSUR,VOLRUN
 211  FORMAT(//,5X,'MODEL TIME = ',I2,' DAYS',//,
     1 5X,'TOTAL RAINFALL            VOLUME = ',F12.2,' CUBIC FEET',/,
     1 5X,'TOTAL EVAPOTRANSPIRATION  VOLUME = ',F12.2,' CUBIC FEET',/,
     1 5X,'TOTAL SUBSURFACE STORAGE  VOLUME = ',F12.2,' CUBIC FEET',/,
     1 5X,'TOTAL DEEP PERCOLATION    VOLUME = ',F12.2,' CUBIC FEET',/,
     1 5X,'TOTAL SURFACE STORAGE     VOLUME = ',F12.2,' CUBIC FEET',/,
     1 5X,'TOTAL SURFACE RUNOFF      VOLUME = ',F12.2,' CUBIC FEET')
      IF(I.NE.1)WRITE(7,212)DTO,DET,DSUB,DPER,DSUR,DRUN
 212  FORMAT(/,
     1 5X,'DAILY RAINFALL            VOLUME = ',F12.2,' CUBIC FEET',/,
     1 5X,'DAILY EVAPOTRANSPIRATION  VOLUME = ',F12.2,' CUBIC FEET',/,
```

```
     1 5X,'DAILY SUBSURFACE STORAGE  VOLUME = ',F12.2,' CUBIC FEET',/,
     1 5X,'DAILY DEEP PERCOLATION    VOLUME = ',F12.2,' CUBIC FEET',/,
     1 5X,'DAILY SURFACE STORAGE     VOLUME = ',F12.2,' CUBIC FEET',/,
     1 5X,'DAILY SURFACE RUNOFF      VOLUME = ',F12.2,' CUBIC FEET',/)
      IF(VOLSUB.EQ.SUBMAX)WRITE(7,213)
 213  FORMAT(5X,'*** SUBSURFACE STORAGE IS FULL ***')
      IF(I.NE.1 .AND. DRUN.EQ.0.)WRITE(7,215)
 215  FORMAT(5X,'*** SURFACE RUNOFF IS ZERO ***')
      IF(DSUB.LT.0.)WRITE(7,216)
 216  FORMAT(5X,'*** SUBSURFACE STORAGE IS USED BY ACTUAL',
     1 ' EVAPOTRANSPIRATION ***')
      IF(VOLSUB.EQ.0.)WRITE(7,217)
 217  FORMAT(5X,'*** SUBSURFACE STORAGE IS EMPTY ***')
      WRITE(7,218)
 218  FORMAT(4X,70('='))
      TTO=VOLTO
      TET=VOLET
      TSUB=VOLSUB
      TSUR=VOLSUR
      TPER=VOLPER
      TRUN=VOLRUN
 100  CONTINUE
C..OUTPUT MONTHLY WATER BUDGET
      TTEMP=TTEMP/REAL(IDAY(JJ))
      WRITE(7,12)IMON(JJ),VOLTO,VOLET,VOLSUR,VOLSUB,VOLPER,VOLRUN,TTEMP
 12   FORMAT(2X,A3,3X,F8.2,1X,F8.2,1X,F8.2,'/',F8.2,4X,F8.2,2X,F8.2,
     1 3X,F5.1,' °F')
      GO TO 1300
 1200 WRITE(7,42)IMON(JJ)
 42   FORMAT(2X,A3,3X,'NO RAINFALL DATA')
C..YEARLY TOTAL SUMMATION
 1300 YTO=YTO+VOLTO
      YET=YET+VOLET
      YRUN=YRUN+VOLRUN
      YTEMP=YTEMP+TTEMP
      YPER=YPER+VOLPER
 1100 CONTINUE
C..OUTPUT YEARLY TOTAL AND AVERAGED VALUES
      WRITE(7,13)YTO,YET,VOLSUR,VOLSUB,YPER,YRUN
 13   FORMAT(/,78('-'),/,2X,'SUM',3X,F8.2,1X,F8.2,1X,F8.2,'/',F8.2,4X,
     1 F8.2,2X,F8.2)
C..YEARLY AVERAGING
      YTO=YTO/12.
      YET=YET/12.
      YRUN=YRUN/12.
      YTEMP=YTEMP/12.
      YPER=YPER/12.
      IF(II.EQ.IYEAR)WRITE(7,14)
 14   FORMAT(2X,'AVG.',/,78('*'),///)
      IF(II.NE.IYEAR)WRITE(7,44)YTO,YET,YPER,YRUN,YTEMP
 44   FORMAT(2X,'AVG.',2X,F8.2,1X,F8.2,22X,F8.2,2X,F8.2,3X,F5.1,' °F',/,
     1 78('*'),///)
 1000 CONTINUE
C
C  END OF SIMULATION
C
 9999 WRITE(7,22)
 22   FORMAT(/,78('='),/,'  ==> END OF WATER BUDGET SIMULATION <==',/)
      WRITE(*,22)
C
      CLOSE (UNIT=7)
```

```
      WRITE(*,41)OFILE
 41   FORMAT('  NOTE: YOUR OUTPUT IS STORED IN [ ',A15,']',///)
C
      GO TO 35
C
      END
C
C  SURFACE-SUBSURFACE WATER BUDGET PROGRAM
C
      SUBROUTINE BUDGET(DEPTH,AET,A,SUBMAX,VOLSUB,VOLPER,
     C                  DINF,DP,DG,VOLET,VOLRUN,PGR)
C..INITIALIZATION OF VARIABLE
      DG=0.
C
C  WATER BUDGET
C
      AVIVOL=VOLSUB
      IF(AVIVOL.NE.0.)GO TO 150
      PG=PGR/100.
      AVIVOL=PG*VOLPER
      VOLPER=VOLPER*(1.-PG)
 150  ETVOL=AET*A
      RAVOL=DEPTH*A
      IF(DEPTH.GE.AET)GO TO 100
      IF(RAVOL+AVIVOL.GT.ETVOL)GO TO 200
C..DEPLETION ALL OF THE AVAILABLE SUBSURFACE STORAGE BY AET
      VOLET=VOLET+RAVOL+AVIVOL
      VOLSUB=0.
      IF(DP.NE.0.)DEPTH=0.
      RETURN
C..DEPLETION PART OF THE AVAILABLE SUBSURFACE STORAGE BY AET
 200  VOLSUB=VOLSUB+RAVOL-ETVOL
      VOLET=VOLET+ETVOL
      IF(DP.NE.0.)DEPTH=0.
      RETURN
C..INFILTARION TO SUBSURFACE STORAGE
 100  IF(DINF.GE.DEPTH-AET)GO TO 110
C..PONDING OR EXCESS SURFACE RUNOFF FROM INFILTARION
      IF(DP.EQ.0.)GO TO 105
      XD=DEPTH-AET-DINF
      DEPTH=XD
      IF(XD.LT.DP)GO TO 105
      DG=XD-DP
      DEPTH=DP
 105  VOLSUB=VOLSUB+DINF*A
      VOLET=VOLET+ETVOL
      IF(DP.EQ.0.)VOLRUN=VOLRUN+(DEPTH-AET-DINF)*A
      GO TO 120
C..TOTAL INFILTARION
 110  VOLSUB=VOLSUB+(DEPTH-AET)*A
      VOLET=VOLET+ETVOL
      IF(DP.NE.0.)DEPTH=0.
C..CHECK DEEP PERCOLATION
 120  IF(VOLSUB.LT.SUBMAX)RETURN
      VOLPER=VOLPER+VOLSUB-SUBMAX
      VOLSUB=SUBMAX
      RETURN
      END
```

Note: TEMP.FIL, RAIN.FIL, and EVAP.FIL files store temperature,
rainfall, and evaporation data, respectively.